
The Great Escape(char)

Louise S. Hadden, Abt Associates Inc., Cambridge, MA

ABSTRACT

SAS® provides programmers with many ways to enhance ODS output in addition to the use of both SAS-supplied
and user-written ODS styles. In-line formatting of titles, footnotes, text fields and table cells and formatting of data
using user-written formats and within the data step can be accomplished by using ODS ESCAPECHAR. A quick
overview of syntax for some of the many possibilities for enhancing ODS output with ODS ESCAPECHAR currently
available in SAS 8.2 and 9.1.3 will be presented, including underlines, pre and post images, special functions, line
feeds and super and subscripts. Some new features available in SAS 9.2 will be discussed.

INTRODUCTION

The use of ODS ESCAPECHAR allows the SAS® programmer to “pass through” formatting commands and
functions into text strings (such as titles, footnotes, and ODS text fields) and variable values (through in-line style
commands, PROC FORMAT, and the data step.) Some formatting commands available include (but are not limited
to!): superscript, subscript, changes within text strings, bold, italic, underline, font color and font face, page x of y,
line feeds, images, and destination-specific elements (such as RTF formatting commands.) Examples of all the
techniques described above will be very briefly demonstrated, for multiple output destinations. For more specific
treatment of ODS ESCAPECHAR, please read papers (and/or attend presentations!) by Cynthia Zender of SAS®,
Scott Huntley of SAS®, Wayne Hester of SAS®, Ray Pass, Lauren Haworth, and Leroy Bessler among others.

As the name implies, ODS ESCAPECHAR is designed to be used in output designated for the Output Delivery
System. The default SAS® listing destination is not able to translate the formatting commands. Below follows a
snippet of a SAS listing from PROC FREQ of values that include formatting commands.

column1 Frequency Percent
--------------------------------------------------------------------
**The probability is 99% or more that your 1 25.00
*The probability is 95% or more that your 1 25.00
^{style [fontsize=6pt] ^{dagger}^{dagger}}The 1 25.00
^{style [fontsize=6pt] ^{dagger}}The 1 25.00

In order to use these special formatting commands, the SAS® programmer must designate a character or characters
which specify that formatting commands follow. Generally, this character is one that would not ordinarily be found in
SAS® data or procedural commands. Examples of commonly used “ESCAPECHARs” are “^”, “~”, and “#”.
Mainframe users will want to be particularly careful in specifying ESCAPECHARs: for example, the “^” is problematic
with some mainframe operating systems and software. The defaults, *ESC* (version 9) and ‘03’x (version 8.2),
contain multiple keystrokes and are used less frequently. It is also possible to specify longer ESCAPECHARs.

The ODS ESCAPECHAR statement should be placed inside the ODS call(s), and the ESCAPECHAR specified must
be the same as in your data and/or style commands. The standard usage of ODS ESCAPECHAR is:

ESCAPECHAR={attribute=value};

However, special formatting commands can be inserted in variables using PROC FORMAT or in the data step. The
variable column1 shown both above and below has formatting commands resulting from a data step command.

The same data shown in the listing above was output to an ODS destination.

ods html file='freq.xls' path=odsout style=styles.journal;
ods ESCAPECHAR='^';
proc freq data=footer;

tables column1;
run;
ods html close;

column1 Frequency Percent

**The probability is 99% or more that

your facility's rate is greater than the

state average.

1 25

*The probability is 95% or more that

your facility's rate is greater than the

state average.

1 25

††The probability is 99% or more that

your facility's rate is greater than the

national average.

1 25

†The probability is 95% or more that

your facility's rate is greater than the

national average.

1 25

For the remainder of this paper, the ESCAPECHARs used will be the tilde (~) or caret (^).

ods ESCAPECHAR='^';
ods ESCAPECHAR='~';

USING ODS ESCAPECHAR IN TITLES AND FOOTNOTES

ODS ESCAPECHAR is most commonly used in titles, footnotes and ODS text fields. The syntax for using ODS
ESCAPECHAR in titles and footnotes is fairly consistent across destinations or tagsets, with some important
exceptions which will be explored below. Style attributes CAN be inserted into titles and footnotes with ODS
ESCAPECHAR, but it is not necessary unless you are “nesting” various attributes, one or more of which require the
use of ODS ESCAPECHAR. In addition to style attributes, text “functions” such as superscripts, subscripts, and
special characters, pre- and post-images, line breaks and non-breaking lines, and RTF or HTML control strings may
be inserted into titles and footnotes using ODS ESCAPECHAR. Finally, special functions such as {pageof} (RTF
only) and {thispage}/{lastpage} (RTF and PDF only) can be inserted into titles and footnotes using ODS
ESCAPECHAR.

The code below demonstrates using ODS ESCAPECHAR and special functions such as superscript, subscript and
dagger; the insertion of pre-images; the insertion of line breaks; and feeding an RTF control string (underline) into
titles. Identical syntax can be used for footnotes. Screenshots of the different destination outputs follow below.

ods pdf file='escapechar_title.pdf';
ods rtf file='escapechar_title.doc' path=odsout style=styles.rtf bodytitle;
ods html file='escapechar_title.xls' path=odsout style=styles.journal;
ods escapechar='^';

proc print data=sashelp.retail (obs=1);
title1 height=3 justify=center italic color=purple 'Title Style Insertion in V9.1 -
ESCAPECHAR Not Needed';
title2 height=2 color=red '^{super SUPER }' color=blue 'COOL ' color=black
'^{sub Title Special Functions (Super, Sub) in V9.1}';

title3 height=4 color=red '^{dagger} Title Special Functions (Dagger) in V9.1';
title4 height=5 '^S={preimage="bonnietiny.jpg"}' "Title Pre-Image in V9.1";
title5 height=2 'Show Line Break in V9.1 ^nHere is an Extra Line –
^nHandy if you want to exceed the 10 title line limit!';

title6 height=3 '^\ul Feed in Underline Command in V9.1 (RTF Only) ^\ul0';
run;

ods _all_ close;

The code below demonstrates using ODS ESCAPECHAR and the special Printer destination specific {pageof} (RTF
only) and {thispage}/{lastpage} (RTF and PDF only) to insert page numbering into titles. Identical syntax can be
used for footnotes. Screenshots of the different destination outputs follow below. NOTE: I have become
accustomed to sending RTF output to the .doc extension out of sheer laziness (when “saving as”, one doesn’t have
to change the extension!) These particular functions will NOT work if you use the .doc extension (I found this out the
hard way!)

ods pdf file='pageof_title_v91.pdf' style=styles.journal;

ods rtf file='pageof_title_v91.rtf' path=odsout style=styles.journal notoc_data;
ods escapechar='^';

proc print data=sashelp.retail (obs=100);
title1 'Page ^{thispage} of ^{lastpage} - for RTF & PDF';
title2 'Page ^{pageof}';
title3 height=4 color=green '^{pageof} Title pageof in V9.1 - RTF only';
title4 height=4 color=blue '^{thispage} of ^{lastpage} Title thispage-lastpage in
V9.1 - RTF and PDF Only';
run;

ods _all_ close;

Results in:

USING ODS ESCAPECHAR IN ODS TEXT FIELDS (PRINTER DESTINATIONS ONLY)

The ODS text command within ODS statements in “Printer” destinations such as RTF and PDF allows the user to
insert lines of text after titles and procedures at will – or without running any procedures at all!

ods rtf text='{\animtext1 This is an example of inserting a text field.}';

Results in:

This is an example of inserting a text field.

ods pdf file='format_ods_text.pdf';
ods escapechar='^';
title 'ODS Text Modification';

ods pdf text = '^S={font_size=18pt just=c font_face=Klingon} This is Klingon';
ods pdf text = '^S={font_size=18pt foreground=blue just=c font_face=Romulan}
This is Romulan';

ods pdf text = '^S={font_size=18pt font_weight=bold foreground=brown just=c
font_face=Federation} This is Federation';

ods pdf text = '^S={font_size=18pt foreground=green just=c font_face=trek}
This is NOT Klingon or Romulan or Federation';

ods pdf text = '^S={font_size=18pt foreground=purple font_style=italic just=c
font_face=arial} This is boring Arial';

ods pdf close;

Results in:

USING ODS ESCAPECHAR IN PROCEDURAL OUTPUT

The syntax for using ODS ESCAPECHAR in procedural output varies by both procedure and destination. Formatting
commands can also be made creating variables containing ODS ESCAPECHAR commands in the data step or
through PROC FORMAT. In-line formatting commands are most commonly used in the reporting procedures (PROC
PRINT, PROC REPORT and PROC TABULATE); however, by passing ODS ESCAPECHAR commands through in
the data step or in PROC FORMAT, text “functions” such as superscripts, subscripts, and special characters, line
breaks and non-breaking lines, and RTF or HTML control strings may be used in the ODS output of other procedures
as well, as shown in the first example using PROC FREQ above. Selected samples will be shown below.

SUPERSCRIPT - RTF

The code below demonstrates using ODS ESCAPECHAR and the superscript function in procedural output.
Subscripts work in a similar way. A screenshot demonstrating the use of the superscript function in RTF output
follows below.

options ps=55 ls=175 errorabend missing=' ' orientation=landscape nocenter nodate
nonumber;

ods rtf file='MMWR_Table5_test.doc' path=odsout style=styles.nmep bodytitle;
ods escapechar = "^";

proc report nowd data=table5 spacing=8
style(report)=[cellpadding=3pt vjust=b]
style(lines)=[just=left font_face=Arial]
style(header)=[just=center font_face=Arial font_weight=bold font_size=8pt]
ls=176;
columns fmtvar ('Main Partner ^{super **} ' ( ('Sex' count1 colper1)

('Unprotected Sex^{super****}' count2 colper2) ) )

('Non-Main Partner ^{super***}' ( ('Sex' count3 colper3)
('Unprotected Sex^{super****}' count4 colper4) ) )

count boldit;
define boldit / display ' ' noprint;
define fmtvar / style(COLUMN)={just=l font_face=Arial

font_size=8pt cellwidth=300 asis=on}
style(HEADER)={just=l font_face=Arial font_weight=bold

font_size=8pt cellwidth=300 asis=on};
. . .

Results in:

SPECIAL FUNCTIONS

The code below demonstrates using ODS ESCAPECHAR and some special functions (dagger and output) in
procedural output. Both data step and PROC FORMAT functionality are used. Screenshots of the different
destination outputs follow below. NOTE: as in the special page numbering functions described above, HTML output
should have an HTML extension, and RTF output should have an RTF extension, for these special functions to work
reliably.

proc format;
value one 1='^{dagger}'

2='^{dagger}^{dagger}';
run;

data one;

one=1;
dagger='^{dagger}';
doubledagger='^{dagger}^{dagger}';
sigma='^{sigma}';

run;

data two;
one=2;
dagger='^{dagger}';
doubledagger='^{dagger}^{dagger}';
sigma='^{sigma}';

run;

data printit;
set one two;
oneunfmted=one;

run;

ods pdf file='dagger_v91.pdf' style=styles.journal;
ods rtf file='dagger_v91.rtf' path=odsout style=styles.journal;
ods html file='dagger_v91.html' path=odsout style=styles.journal;

ods escapechar='^';

proc print data=printit noobs;
var oneunfmted one dagger doubledagger sigma;

format one one.;
title1 color=red bcolor=black 'Test Special Functions - V9.1';
run;

ods _all_ close;

LINE BREAKS AND OTHER SPECIAL CODES

Below follows a screenshot of an Excel spreadsheet with 3 fields on item importance. To produce 15,000+ reports
on nursing home facilities, this spreadsheet is imported into SAS® and the three items concatenated into a single
field, which is then used to generate a item-level page for each facility and item. You will notice ^n^n in the second 2
fields on item importance. This generates 2 line breaks (essentially starts a new paragraph) within the variable.

data printimp;
length importance $ 1032;
set itemtemp;
intro="Importance of "||trim(item_name)||" as an MDS Accuracy Trigger:";
importance=catx(' ',item_importance_1,item_importance_2,item_importance_3);

run;

data _null_;
set itemtemp;
call symput("callitem",trim(item_name)||' ('||trim(propcase(item_short))||')');
call symput("calldef",item_def);

run;

data _null_;
set printimp;
call symput("intro",intro);

run;

. . .

ods proclabel="Importance";
ods escapechar='^';

proc report nowd data=printimp spacing=8
style(report)=[cellpadding=3pt vjust=b]
style(lines)=[just=left font_face=Arial]
style(header)=[just=left font_face=Arial font_weight=bold font_style=italic

font_size=10pt]
ls=176 contents="Importance";
columns ("&intro" importance);
define importance / ' ' style(COLUMN)={just=l font_face=Arial font_style=italic

font_size=8pt cellwidth=1000}
;

title1 ;

The code snippets shown above result in PDF pages that include the following:

HTML AND RTF CONTROL STRINGS

We have seen above that we can insert line feeds with ODS ESCAPECHAR, into all three destination “families.”
There may be other similar “events” that we would like to have in our output, such as tabs. These events are often
destination-specific. In addition, we may want to “stack up” or nest these events and style attributes. ODS
ESCAPECHAR R allows this functionality. Note that insertions done with ODS ESCAPECHAR R are non-printing in
list output while insertions done with ESCAPECHAR N do print in list output. This may affect the length of character
variables including ESCAPECHAR R sequences in data sets and data set size – they will be shorter. On the other
hand, you have to know your control strings pretty well! The code and output samples below show how to insert line
feeds both using ESCAPECHAR N and ESCAPECHAR R, and how to add style attributes.

proc format;
value xit 1="Green"

2="Red";
value xtoo 1="1 ^n Yes" /* ODS ESCAPECHAR N */

2="2 ^n No";
value $ fit "1"-<"2"="Green"

"2"-<"3"="Red";
run;

data one;
length feed $ 132;
xit=1;
xtoo=2;
feed=cat("1 ","^R/HTML'<BR/>' ^R/RTF'\line'"," Yes"); /* ODS ESCAPECHAR R */

run;

data two;
length feed $ 132;
xit=2;
xtoo=1;
feed=cat("2 ","^R/HTML'<BR/>' ^R/RTF'\line'"," No"); /* ODS ESCAPECHAR R */

run;

/* Note that the string defining the variable FEED includes a line feed
for both HTML and RTF */

/* The variable FEED will not print out as expected in PDF */

data printit;
set one two;

run;

ods pdf file='style_format_v91.pdf' style=styles.journal;
ods rtf file='style_format_v91.rtf' path=odsout style=styles.journal bodytitle;
ods html file='style_format_v91.html' path=odsout style=styles.journal;
ods escapechar='^';

proc print data=printit label noobs;
var xit / style={just=l foreground=xit.};
var xtoo;
var feed / style={just=l foreground=$fit.};

format xtoo xtoo. ;
title1 ‘Line Feeds’;
run;

ods _all_ close;

As can be seen, the variable FEED is shown both color-coded and with a line feed in the HTML and RTF
destinations. The HTML and RTF control strings embedded in the FEED variable do not print in the PDF destination.

COMING ODS ESCAPECHAR ATTRACTIONS IN VERSION 9.2

UNICODE CHARACTERS

The following program creates a data set with a number of variables containing Unicode symbols using the Unicode
style function, which is new in V9.2. These variables are also labeled with Unicode symbols and the output is titled
with Unicode symbols. As can be seen in the graphic below, these Unicode symbols are translated by ODS into
graphic characters. This allows the addition of useful characters such as Greek letters, etc. without changing fonts.
The syntax is ^{Unicode xxxx} where ^ is the specified ODS ESCAPECHAR and xxxx is the 4-digit Unicode value.
To look up Unicode values on a system running Windows XP, open Start -> Programs -> Accessories -> System
Tools -> Character Map. For example, I chose Thorndale AMT, which is used by SAS® V9.2. When a character on
the screen is selected and/or hovered over, the Unicode value is shown on the bottom of the screen (in this case the
Black Club Suit is Unicode value 2663.)

data escapechar_v92;
smiley='^{style [fontsize=14pt] ^{unicode 263B} black smiley}';
sunny='^{style [fontsize=14pt] ^{unicode 263C} sunny}';
girly='^{style [fontsize=14pt color=pink fontweight=bold] ^{unicode 2640} girly}';
manly='^{style [fontsize=14pt color=lightblue] ^{unicode 2642} manly}';
heart='^{style [color=red fontsize=14pt] ^{unicode 2665} heart}';
doubledagger='^{style [fontsize=14pt color=blue]

^{unicode 2021} doubledeckerdagger}';
label smiley='^{unicode 263b} Smile,^n Smile,^n Smile!'

sunny='^{unicode 263c} Keeping ^nthe Clouds Away'
girly='^{style [color=pink] ^{unicode 2640} I like Pink!}'
manly='^{style [color=lightblue] ^{unicode 2642} Me Tarzan}'
heart='^{style [color=red] I ^{unicode 2665} Tagsets}'
doubledagger='^{unicode 2020} If you prick me,^n do I not bleed?';

run;

ods pdf file='unicode.pdf';
ods rtf file='unicode.doc' path=odsout style=styles.astronomy bodytitle;
ods html file='unicode.xls' path=odsout style=styles.journal;
ods escapechar='^';
proc print data=escapechar_v92 label noobs;

var smiley sunny girly manly heart doubledagger;
title1 height=3 '^{style [just=c color=purple] ^{unicode beta} Test Unicode in V9.2}';
title2 height=2 color=red '^{super SUPER}' color=blue 'man';
run;

ods _all_ close;

TEXT DECORATION STYLE ATTRIBUTE

TextDecoration is a new style element in V9.2. It incorporates such items as underlines, overlines and strike-
throughs. The code and samples below demonstrate underlining both using TextDecoration and using the
SAS/GRAPH “UNDERLIN” syntax (available in earlier versions), in titles, labels and variables. Overlines and strike-
throughs work in a similar fashion (TextDecoration only!)

data printit;
set sashelp.retail (obs=5);
underline_test='^S={textdecoration=underline}Underline';

label underline_test='Test of^n ^S={textdecoration=underline} Underline';
run;

ods pdf file='text_decoration_v92.pdf' notoc;
ods rtf file='text_decoration_v92.doc' path=odsout style=styles.journal bodytitle;
ods html file='text_decoration_v92.html' path=odsout style=styles.journal;
ods escapechar='^';

proc print data=printit label noobs;

title1 height=3 justify=center italic color=purple underlin=1
'Underlining the OLD Way (SAS/GRAPH Underlin=1)';

title2 height=3 justify=center italic color=purple underlin=2
'Underlining the OLD Way (SAS/GRAPH Underlin=2)';

title3 height=3 justify=center italic color=purple underlin=3
'Underlining the OLD Way (SAS/GRAPH Underlin=3)';

title4 '^S={textdecoration=line_through font_size=16pt color=red}

Coders' CornerNESUG 2007



13

Line Through';
title5 '^S={textdecoration=overline color=blue font_size=14pt}
Line Over';

title6 '^S={textdecoration=underline color=green font_size=18pt}
Underline the NEW Way';

run;

ods _all_ close;

CONCLUSION

SAS® provides us with many tools to customize ODS output. ODS ESCAPECHAR allows SAS® programmers
greater control and flexibility in producing “camera ready” output from SAS® procedures, by enhancing titles and
footnotes, ODS text fields, and procedural output.

